import graph;
import markers;
texpreamble("\usepackage[load-configurations=abbreviations,detect-all=true]{siunitx}
\usepackage{amsmath}
");

size(5*72, 0);
pair Vca = (1,2);

real alpha = 60.0/180.0*pi;
real e1L = 2.0;
real e2L = 1.0;
pair e1v = (e1L, 0);
pair e2v = (e2L*cos(alpha), e2L*sin(alpha));
pair Vv = (e1v*Vca.x + e2v*Vca.y);

real omega1L = 1.0 / (e1L * cos(pi/2 - alpha));
pair omega1v = (omega1L * cos(pi/2 - alpha), - omega1L * sin(pi/2 - alpha));
real omega2L = 1.0 / (e2L * cos(pi/2 - alpha));
pair omega2v = (0, omega2L);

draw((0,0)--e1v, Arrow);
label("$\mathbf{e}_1$", e1v, SE);
draw((0,0)--e2v, Arrow);
label("$\mathbf{e}_2$", e2v, NW);
draw((0,0)--Vv, Arrow);
label("$\mathbf{v}=\begin{cases}u^\mu = (1,2)\\u_\mu=(6,3)\end{cases}$", Vv, E);
draw(e1v--Vv--e2v*Vca.y--e2v, dashed);

draw((0,0)--omega1v, Arrow);
label("$\mathbf{\omega}^1$", omega1v, S);
draw((0,0)--omega2v, Arrow);
label("$\mathbf{\omega}^2$", omega2v, NW);

draw(omega1v--(Vv.x,-Vv.x * tan(pi/2-alpha))--Vv--(0,Vv.y+Vv.x * tan(pi/2-alpha))--omega2v, dashed);

markangle("$\alpha$", e1v, (0.0,0.0), e2v);
